Pharmacology Biochemistry & Behavior, Vol. 15, pp. 135–140, 1981. Printed in the U.S.A.

Use of a Programmable Protocol Timer and Data Logger in the Monitoring of Animal Behavior

RODNEY B. MURRAY, DEBRA E. GMEREK, ALAN COWAN AND RONALD J. TALLARIDA

Department of Pharmacology, Temple University School of Medicine
Philadelphia, PA 19140

Received 8 November 1980

MURRAY, R. B., D. E. GMEREK, A. COWAN AND R. J. TALLARIDA. *Use of a programmable protocol timer and data logger in the monitoring of animal behavior.* PHARMAC. BIOCHEM. BEHAV. 15(1) 135–140, 1981.—The quantitative assessment of animal behavior by continuous or intermittent observation often requires much time and intense concentration. We have developed an accurate and convenient system which allows one observer to monitor up to four animals and record two types of behavior simultaneously. The system uses an inexpensive portable microcomputer including keyboard, video monitor, and cassette recorder. A program written in BASIC generates timed visible and audible cues, and tabulates data entered through the keyboard. The total number of occurrences of a particular behavior (e.g., "wet dog" shakes) and the frequency of a second behavior (e.g., grooming) may be measured. Frequency data, based on intermittent observations, is processed for histogram display. The results may be copied from the video display or recorded on tape for further statistical analysis. The system has provided a convenient way to time experiments and collect data on drug-induced behavior in over 800 rats.

Animal behavior Microcomputer Protocol timing Data aquisition

QUANTITATIVE assessment of animal behavior can be a major problem in psychopharmacological experiments. Some behaviors are easily measured. For example, the "wet-dog" shaking, jumping, and writhing [2] that occur in rats undergoing withdrawal from morphine are behaviors that can be counted without difficulty. Other behaviors are not as easily quantified; for example, the frequency or amount of time an animal spends grooming [5], gnawing [7], and biting [3], or the extent of social interaction [4], or aggressiveness [1]. Recording the amount of time that an animal spends doing a specific activity is time consuming and may require an observer for each animal. Intermittent monitoring allows the observation of more than one animal at a time, but requires constant clock-watching and high concentration. The intensity of a behavior can often be scored using a rating scale. Note, however, that such scales are defined in an arbitrary way and they are not always translatable between different laboratories.

In this paper, we describe a system whereby two behaviors can be simultaneously monitored by one observer in as many as four animals. We measure the frequency of grooming and the number of "wet-dog" shakes in rats [6], but any other two behaviors can be followed. One of the behaviors is measured intermittently and the other is counted over the entire session. No clock-watching is necessary and, at the end of the observation period, the data is in a form ready for analysis. Results, in terms of the frequency of a behavior, allow comparison of data between laboratories and between compounds producing similar behaviors.

METHOD

Hardware

The microcomputer used is a TRS-80 Model I (Radio Shack) with Level II BASIC and 16K bytes of RAM. Similar microcomputers may also be used, e.g., the TRS-80 Model III (the software-compatible successor to the Model I, which is no longer being manufactured). This system includes a keyboard, video display and cassette recorder. The program is loaded and data are stored on magnetic tape using the audio tape recorder supplied with the computer. The only additional hardware required is an external amplifier-speaker which is connected to the cassette output port.

The entire microcomputer, with video monitor and tape recorder, is mounted on a portable typewriter table which may be rolled to the location of the experiment. The operator arranges the computer and observation cages as shown in Fig. 1.

Software

The program, called "OBSERVE" [8], is written in BASIC, and displayed in Fig. 2. OBSERVE generates timed visible and audible cues to aid the scoring of drug-induced behaviors in a group that may be as large as four animals. In a typical experiment, the observer monitors two behaviors simultaneously. The protocol calls for counting: (1) the total number of occurrences of a certain behavior (e.g., "wet-dog" shaking), and (2) the frequency of a second behavior



FIG. 1. The experimental set-up; note video graphic "flag" used to prompt the user to observe animal 3. The screen shows cumulative total of behavior 1 (middle row of numbers) and occurrence of behavior 2 in the current observation period (bottom row).

```
1000 '                  OBSERVE/BAS VERSION 1.0
1010 '                                          INITIALIZATION
1020 CLEAR 2000
1030 CLS:PRINT
1040 PRINT TAB(10) "<  O B S E R V E  >   BY R.B. MURRAY"
1050 DEFINT A-Z
1060 DIM N(23),NA$(23)
1070 G$=STRING$(4,CHR$(191))
1080 W$=STRING$(4,CHR$(128))
1090 '                                          USER'S INPUT
1100 PRINT:INPUT"FILE NAME (UP TO 8 CHARACTERS) ";DF$
1110 IF LEN(DF$)<1 OR LEN(DF$)>8 THEN RUN
1120 INPUT"DURATION OF OBSERVATION PERIOD (SEC)";DP
1130 INPUT"ENTER NUMBER OF OBSERVATION PERIODS ";NP
1140 INPUT"ENTER NUMBER OF ANIMALS (1-4) ";NR
1150 IF NR<1 OR NR>4 THEN 1140
1160 DIM GC(NR,121),DA(NR,23)
1170 S=NR*DP*NP:M=S/60
1180 PRINT"EXPERIMENT WILL LAST";M;"MIN AND";S-(60*M);"SEC."
1190 INPUT"PRESS ENTER TO START TEST ";K$
1200 CLS:PRINT@128,CHR$(23);
1210 TB=36/NR
1220 FOR R=0 TO NR-1:PRINTTAB(R*TB);"#";R+1;:NEXT R
1230 F$="###%        %###%        %###"
1240 '                                          BEGIN PROTOCOL
1250 FOR P=1 TO NP:'                               PERIOD LOOP
1260   PRINT@452,"PERIOD #";P
1270   FOR R=0 TO NR-1:'                           ANIMAL LOOP
1280     I$=""
1290     PRINT@0,"";
1300     PRINTTAB(R*TB);G$;
1310     FOR Q=0 TO R:'                            BUZZER
1320       FOR BZ=1 TO 10:OUT 255,1:OUT 255,2:NEXT BZ
1330       FOR BZ=1 TO 10:NEXT BZ
1340     NEXT Q:PRINTCHR$(23);
1350     FOR T=1 TO DP*2:'                         TIMING LOOP
1360       PRINT @478,"";:PRINT"SECONDS:";
1370       PRINT USING "##";T/2
1380       FOR U=1 TO 25:NEXT U:REM <<== ADJUST TIMING HERE ==<<
1390       I$=INKEY$
1400       IF I$="U" THEN SC(0)=SC(0)+1
1410       IF I$="I" THEN SC(1)=SC(1)+1
1420       IF I$="O" THEN SC(2)=SC(2)+1
1430       IF I$="P" THEN SC(3)=SC(3)+1
1440       IF I$=" " THEN GC(R,P)=1
1450       PRINT@192,"";
1460       FOR R1=0 TO NR-1
1470         PRINTTAB(R1*TB);:PRINT USING F$;SC(R1);
1480       NEXT R1
1490       PRINT@256,"";
1500       FOR R1=0 TO NR-1
1510         PRINTTAB(R1*TB);:PRINT USING F$;GC(R1,P);
1520       NEXT R1
1530     NEXT T
1540   PRINT@0,"";
1550   PRINTTAB(R*TB);W$;
1560   NEXT R
1570 NEXT P:'                                   END PROTOCOL
1580 '                                          PRINT REPORT
1590 CLS:BF=0:C=0
1600 FOR R=0 TO NR-1
1610   PRINT"ANIMAL #";R+1;"  GROOMING: ";
1620   C=0:BF=0:G(R)=0
1630   FOR P=1 TO NP
1640     G(R)=G(R)+GC(R,P)
1650     PRINT GC(R,P);
1660     IF GC(R,P)<>1 THEN 1700
1670     IF BF=0 THEN C=0:BF=1:GOTO 1700
1680     IF C>19 THEN C=20
1690     IF BF=1 THEN DA(R,C+2)=DA(R,C+2)+1:C=0
1700     C=C+1
1710   NEXT P:PRINT
1720   PRINT"  TOTAL GROOMING = ";G(R),"TOTAL SHAKING = ";SC(R)
1730 NEXT R
1740 '                                          FORM DATA ARRAYS
1750 IF NP<21 THEN K=1+NP ELSE K=22
1760 JM=5:'     K = # OF BINS+2   JM = MAX # ANIMALS+1
1770 N(1)=NR: NA$(1)="TOTAL SHAKES"
1780 FOR X=0TONR-1: DA(X,1)=SC(X): NEXT X
1790 N(2)=NR: NA$(2)="TOTAL GROOMING"
1800 FOR X=0TONR-1: DA(X,2)=G(X): NEXT X
1810 FOR N=3 TO K
1820   N(N)=NR:T$=STR$(N-2):NA$(N)="B"+RIGHT$(T$,LEN(T$)-1)
1830 NEXT N
1840 '                                          COMPRESS DATA
1850 PRINT:PRINT"HISTOGRAM DATA (";K-2;" BINS ):"
1860 D$="":E$="/":N=1
1870 D$=DF$+E$+STR$(K)+E$+STR$(JM):GOSUB 2010
1880 FOR J=1 TO K
1890   D$=STR$(N(J))+E$+NA$(J):GOSUB 2010
1900   FOR I=0 TO N(J)-1
1910     D$=STR$(DA(I,J)):GOSUB 2010
1920   NEXT I
1930 NEXT J
1940 FOR J=3 TO K:PRINT NA$(J);
1950  FOR I=0 TO N(J)-1:PRINT DA(I,J);:NEXT I:PRINT"",
1960 NEXT J
1970 GOSUB 2060: GOSUB 2060
1980 END
1990 '                                          SUBROUTINES
2000 '                                             ENCODE STRING
2010 DL=LEN(D$)+1:TL=LEN(T$(N))
2020 IF DL+TL>230 THEN N=N+1
2030 T$(N)=T$(N)+D$+E$:D$=""
2040 RETURN
2050 '                                             RECORD DATA
2060 PRINT
2070 INPUT"PRESS <ENTER> WHEN TAPE RECORDER IS READY";I$
2080 PRINT"*** WRITING FILE TO TAPE ***"
2090 PRINT#-1,N
2100 FOR X=1 TO N:PRINT#-1,T$(X):NEXT X
2110 RETURN
2120 END
```

FIG. 2. Listing of OBSERVE program. The timing loop is calibrated for observing four animals. The length of this loop ("25" in line number 1380) must be adjusted for each computer.

(e.g., grooming). Timing is accomplished by means of program loops as this microcomputer configuration does not include a real-time clock. The length of the timing loop (the number "25" on line 1380) must be calibrated to account for timing differences between individual machines. Timing accuracy (and cost) may be increased by adding real-time clock hardware, although for experiments of 0.5–1 hr duration, the accuracy of program loop timing is sufficient (±5 sec/hr).

Operating Procedure

Once the program is loaded, the operator is prompted to enter the parameters of the particular experimental protocol by messages displayed on the video monitor (Fig. 3A). The program times the observation periods for individual animals and prompts the user both visibly and audibly.

The occurrence of behavior 1 (shaking) is recorded when the operator depresses specific keys on the keyboard. The incidence of shaking in animals numbered 1 through 4 is tabulated by depressing keys "U," "I," "O," and "P," respectively. Behavior 1 may be scored whenever it occurs during the experimental period. The score for this behavior is displayed continuously on the video monitor (Figs. 1 and 3B).

Behavior 2 (grooming) is scored as an all-or-none effect during intermittent periods of observation. The video



FIG. 3. Simulation of program messages which appear on the video screen. Panel A shows prompting questions used to initialize program for a specific experimental protocol. Panel B illustrates the video display during the testing (see legend for Fig. 1). The top of panel C displays the results of the experiment: the total number of occurrences of each behavior and a string of 1's and 0's indicate each occurrence of behavior 2 (grooming). These frequency data for behavior 2 are accumulated into 9 "bins" according to their inter-event-interval for histogram display (bottom of panel C).



FIG. 4. Histogram display of grooming frequency. Frequency data are based on 5 sec observation periods every 15 sec. The number of occurrences of grooming (ordinate) is plotted against inter-grooming-interval (abscissa). Histogram A is derived from two rats that received 3 μl of $ACTH_{1-24}$ (1 mg/ml) ICV and saline SC. Histogram B is derived from two rats which received the same dose of $ACTH_{1-24}$ ICV and 1 mg/kg of naloxone SC.

monitor displays a large graphic "flag" under the appropriate animal (see Fig. 1) and "beeps" a number of times corresponding to the animal's identification number. The operator observes each animal in sequence for a selected time period, usually 5 sec. If four animals are being monitored there is a 15 sec lapse between observation periods. Pressing the space bar on the keyboard indicates that a particular animal has shown behavior 2 (regardless of duration) during the observation period. Depressing the key a second time during a given observation period has no effect.

After the experiment, the video monitor displays the total counts for both behaviors as well as the frequency data for behavior 2. Figure 3C shows a sequence of 0's and 1's indicating the occurrence of behavior 2. These data are accummulated into 9 "bins" (labeled B1 through B9 at the bottom of Fig. 3C) according to the inter-event-interval for histogram display. A representative histogram is shown in Fig. 4.

DISCUSSION

The proliferation of low-cost microcomputers in the laboratory, usually for statistical analysis, has led in this case to a novel use in psychopharmacology. Tasks which used to require specialized hardware may now be accomplished with less effort and expense by using software. The program has been run during the past 9 months to tabulate data for over 800 animals. The system has provided a means for conveniently timing protocols and recording data in studies of rat behavior. In addition, use of the microcomputer offers a novel way of prompting the observer to perform what are often tedious tasks in a more pleasant and productive manner.

ACKNOWLEDGEMENTS

This work was supported by Grant DA 02322 from NIDA and Grant BSRG S07 RR05417 from the Division of Research Resources, NIH.

REFERENCES

1. Avis, H. A. and H. V. S. Pecke. Morphine withdrawal induced in the Syrian hamster. *Pharmac. Biochem. Behav.* **11:** 11–15, 1979.
2. Bläsig, J., A. Herz, K. Reinhold and S. Zieglgänsberger. Development of physical dependence on morphine in respect to time and dosage and quantification of the precipitated withdrawal syndrome in rats. *Psychopharmacologia* **33:** 19–38, 1973.
3. Costall, B. and R. J. Naylor. On the mode of action of apomorphine. *Eur. J. Pharmac.* **21:** 350–361, 1973.
4. File, S. E. and J. R. G. Hyde. A test of anxiety that distinguishes between the actions of benzodiazepines and those of other minor tranquilisers and of stimulants. *Pharmac. Biochem. Behav.* **11:** 65–69, 1979.
5. Gispen, W. H., V. M. Wiegant, H. M. Greven and D. deWied. The induction of excessive grooming in the rat by intraventricular application of peptides derived from ACTH: structure-activity studies. *Life Sci.* **17:** 645–652, 1975.
6. Gmerek, D. E. and A. Cowan. RX 336-M and $ACTH_{1-24}$: stimulants of grooming and shaking in rats. *Soc. Neurosci. Abstr.* **6:** 701, 1980.
7. Langwinski, R. and J. Niedzielski. Narcotic analgesics and stereotyped behavior in mice. *Archs Pharmac.* **312:** 225–227, 1980.
8. Murray, R. B. and R. J. Tallarida. Protocol timing and data acquisition by microcomputer. *Soc. Neurosci. Abstr.* **6:** 423, 1980.